Example nº 19
Cable – Stayed Bridge Construction
(Example 4)

CivilFEM Manual of Advanced Examples Example nº 19 – Table of Contents

19 EXAMPLE Nº 19: CABLE – STAYED BRIDGE CONSTRUCTION (Example 4) 1
19.1 AIM ... 1
19.2 DESCRIPTION OF THE EXAMPLE ... 1
19.3 RESULTS TO BE OBTAINED ... 2
19.4 CALCULATION LOG ... 3
19.4.1 Introduction ... 3
19.4.2 Model generation ... 4
19.4.3 Solution 1 ... 6
19.4.4 Postprocess 1 ... 7
19.4.5 Solution 2 ... 8
19.4.6 Postprocess 2 ... 8
19.4.7 Postprocess 3 (POST26) ... 9
19.4.8 Macros for postprocessing ... 9
19.5 RESULTS ... 11
19.5.1 Bending moment (MZ) in different phases of the construction 11
19.5.2 Other results in the final state of the structure ... 14
19.5.3 Envelope of bending moment (MZ) ... 16
19.5.4 Time history results (POST26) ... 17

19 EXAMPLE Nº 19: CABLE – STAYED BRIDGE CONSTRUCTION (Example 4)

19.1 AIM

The aim of this example is to demonstrate how to determine the initial cable forces in a cable-stayed bridge. The initial condition for the cable force calculation is that the desired shape of the deck is horizontal (zero vertical deflection). In the second part the backward construction process analysis is performed on the model.

19.2 DESCRIPTION OF THE EXAMPLE

The figure below is a sketch of the bridge model.



<u>DATA:</u>

GEOMETRY

HTAbove = 45 m	Height of tower above deck level.
ACable = 20cm^2	Cable area.
NCables = 9	Number of cables.
XDistCable = 10 m	Distance between cables at deck level.
CONCRETE	HA-40 (EHE)
CABLE STEEL	Y1860S7 (EHE)

ADDITIONAL DEAD LOAD 10 kN/m

CALCULATION STEPS:

1. Cable forces are calculated in the final state of the bridge.
    - The deformations of the deck due to the self weight of the structure and additional dead load is calculated.
    - Every cable is prestressed by unit change in the length (-1 m) one by one and the deformations of the deck is calculated for each case.
    - A system of linear equations is assembled: Kx = U.
        - The response of the structure to the unit cable forces are the elements of the coefficient matrix (K).
        - The deformations (UY of the deck) are the elements of the vector on the right-hand side (U).
    - The system of linear equations is solved $\rightarrow$ x is determined.
        - The elements of the x vector are the actual cable forces (i.e. cable shortening) to achieve the desired shape of the structure.
    - The x vector is applied as cable forces (i.e. cable shortening) and the structure is solved again.
2. Backward analysis of the construction process ('demolishing' the bridge step by step).
    - Remove additional dead load.
    - Remove rotational boundary condition in the middle of the bridge.
    - Remove the last section of the deck (located after the last cable).
    - Loop over the rest of the structure.
        - Remove the cable.
        - Remove segment of the deck.

19.3 RESULTS TO BE OBTAINED

In each construction phase:

1. Deflection of the structure.
2. Bending moments diagram.
3. Axial forces diagram.
4. Cable force (shortening) variation.
5. Envelope of bending moments due to the construction.

19.4 CALCULATION LOG

19.4.1 Introduction

A CivilFEM box bridge section is used to model the bridge deck, therefore the Bridges Module needs to be activated.

For the postprocessing of the results a series of macros are created (the macro file, Example19_macros.inp, needs to be placed in the working directory). For each macro a button is placed in the top toolbar:

START	- Go to the first phase of the construction (last load step).
+1	- Go to the next phase.
-1	- Go one phase back.
END	- Go to the final phase of the construction (first load step).
UY	- Vertical deflection of the structure.
MZ	- Bending moment about the axis Z.
FX	- Axial force in the deck.
CF	- Cable forces.
CST	- Applied cable shortening.
CF_GRAPH	- Graph showing the variation in the cable forces during the phases of the construction.
ENV_MZ-	- Envelope of negative bending moments.
ENV_MZ+	- Envelope of positive bending moments.

For the cables LINK11 elements are used. This element allows the application of shortening as an external load.

When the meshing is done, components of elements that correspond to each construction phase are created, both for the deck and the cables.

In the first solution phase the deflection of the structure due to its self weight and the additional dead load are calculated, and the response of the structure to the unit cable forces (shortening) in each cable is obtained.

In the first postprocessing phase the Kx=U equation is created and solved.

In the second solution phase gravity is applied, and the additional dead load and the calculated cable forces (shortening) in each cable are placed on the model. This is the final state of the bridge and the start state of the backward construction process analysis. After removing the additional dead load, the symmetry boundary condition, a step-by-step demolition of the bridge is done using the Birth & Death of Elements feature.

In the second postprocessing phase the results in the structure are checked for each phase of the construction. To see the cable forces and cable shortening two element tables must be created, since these results of LINK11 elements are not accessible otherwise. To create envelopes of the bending moments the Combination module of CivilFEM is employed.

In the third postprocessing phase the history of the cable force variation is plotted. This needs to be done in the POST26 Time History Postprocessor.

The buttons in the top toolbar can be used for quick access to the results. Using the buttons to navigate between the load steps the macro not only reads the results but also automatically selects the elements that correspond to the actual construction phase.

19.4.2 Model generation

```
  FINISH
  ~CFACTIV,NLBR,Y
  ~CFCLEAR,,1
  /TITLE,Cable-Stayed Bridge Construction Process Example4 (Backward)

  !use library of macros
  *ULIB,Example19_macros,inp

  *ABBR,START,*USE,TO_START
  *ABBR,+1,*USE,NEXT
  *ABBR,-1,*USE,BACK
  *ABBR,END,*USE,TO_END
  *ABBR,UY,*USE,UY
  *ABBR,MZ,*USE,MOMENTZ
  *ABBR,FX,*USE,FORCEX
  *ABBR,CF,*USE,CFORCE,IStep
  *ABBR,CST,*USE,CSTROKE,IStep
  *ABBR,CF_GRAPH,*USE,CFORCE_GRAPH
  *ABBR,ENV_MZ-,*USE,ENVELOPE,1
  *ABBR,ENV_MZ+,*USE,ENVELOPE,2


! Setup
! ----------------------------------------------------
    ~UNITS,SI


! Parameters
! ----------------------------------------------------
HTAbove=45                      ! height of tower above deck level
ACable=20E-4                    ! cable area
NCables=9                       ! number of cables
XDistCable=10                   ! distance between cables at deck level

! Element divison numbers
DeckEDiv=6                      !deck

! Number of construction phases
NPhases=NCables + 1            !+1 to close the bridge

! additional dead load
GsLoad=10e3                     !10 kN/m

! View options
  /VIEW,1,0,0,1
```

```
  /VUP,1,Y
  /ESHAPE,1


! ---------------------------------------------------
! PREPROCESSING
! ---------------------------------------------------
/PREP7

! Element types
! ---------------------------------------------------
  ET,1,BEAM44               !stiffening girder
  ET,3,LINK11               !cables


! Materials
! ---------------------------------------------------
  ~CFMP,1,LIB,CONCRETE,EHE,HA-40      ! tower, stiffening girder
  ~CFMP,2,LIB,REINF,EHE,B 500 S
  ~CFMP,3,LIB,PREST,EHE,Y1860S7       ! cables


! Bridge cross-sections
! ---------------------------------------------------
  ~BRSBOX,1,1,4 ,1.5,2 ,0.2,0.2,0.2,2
  ~BRSMDF,1,BOX,KSYM,,1
  ~BRSMDF,1,BOX,WEB,RATS,0.3
  ~BRSMDF,1,BOX,WEB,SLPS,0.15
  ~BRSMDF,1,BOX,WEB,THICK,0.3,2
  ~BRSMDF,1,BOX,WEB,THICK,0.4,3
  ~BRSMDF,1,BOX,WEB,SLOPE,0.5,1
  ~BRSMDF,1,BOX,WEB,DEPTH,1.4,1

! Cross-section
! ---------------------------------------------------
  !create cross-section from the bridge cross-section
  ~BRSTOCS,1,1,0,0,0,0                              !stiffening girder


! Beam properties (Real constants)
! ---------------------------------------------------
  ~BMSHPRO,1,BEAM,1,1,,,44,1,0      !stiffening girder


! Solid model (KPs and Lines)
! ---------------------------------------------------

! Keypoints
    K,1,0,HTAbove,0                   !top of tower
    K,2,0,0,0                         !deck starting point

! Lines

    ! deck and cables
    PrevKP=2                          !startinh KP of the deck (previous KP)
    *DO,I,1,NCables
        K,,I*XDistCable,0,0
        NewKP=_RETURN                 !new KP of the deck
        L,PrevKP,NewKP                !line of deck
        CM,L_Deck_%I%,LINE
        LSEL,NONE
        L,1,NewKP                     !line of cable
        CM,L_Cable_%I%,LINE
        LSEL,NONE
        PrevKP=NewKP
    *ENDDO

    !closing segement of the deck
```

```
    K,,(NCables+0.5)*XDistCable,0,0
    NewKP=_RETURN                  !new KP of the deck
    L,PrevKP,NewKP                 !line of deck
    CM,L_Deck_%NPhases%,LINE

ALLSEL

! Meshing
! ---------------------------------------------------
! Line divisions for meshing
    LSEL,S,LOC,Y,0
    LESIZE,ALL,,,DeckEDiv             !stiffening girder
    LSEL,INVE
    LESIZE,ALL,,,1                    !cables

ALLSEL

! stiffening girder
TYPE,1
MAT,1
REAL,1
*DO,I,1,NPhases
    CMSEL,S,L_Deck_%I%
    LMESH,ALL
    CM,E_Deck_%I%,ELEM
    ESEL,NONE
*ENDDO

! cables
TYPE,3
MAT,3
*GET,ECable,EX,3

*DIM,CableData,,NCables,1          !to store cable data: ElemID
*DO,I,1,NCables
    CMSEL,S,L_Cable_%I%
    LSUM
    *GET,LCable,LINE,,LENG
    KCable=ECable*ACable/LCable    !cable stiffness K=EA/L
    R,I+2,KCable                   !real constant - different for each cable
    REAL,I+2
    LMESH,ALL
    CM,E_Cable_%I%,ELEM
    !fill data vector
    *GET,CableData(I,1),ELEM, ,NUM,MAX
    ESEL,NONE
*ENDDO

!component of all the deck elements
*DO,I,1,NPhases
    CMSEL,A,E_Deck_%I%
*ENDDO
CM,E_Deck_All,ELEM

!list of nodes where the cables connects to the deck
*DIM,DeckNodes,,NCables
*DO,I,1,NCables
    DeckNodes(I)=NODE(I*XDistCable,0,0)
*ENDDO

ALLSEL
```

19.4.3 Solution 1

```
! SOLUTION
!---------------------------------------------------------
! 1. small displacement analysis for cable force calculation
!    in the final stage of the bridge

/SOLU
    ANTYPE,STATIC
    OUTRES,ALL,ALL

! Boundary condition
NSEL,S,LOC,X,0
NSEL,A,LOC,Y,HTAbove
D,ALL,ALL

!symmetry condition
NSEL,S,LOC,X,(NCables+0.5)*XDistCable
D,ALL, , , , , ,ROTZ,

!additional dead load (Gs)
CMSEL,S,E_Deck_All
SFBEAM,ALL,2,PRES,GsLoad

ALLSEL
ACEL,,9.81,
SOLVE       !LS1

*DO,I,1,NCables
    ACEL,,                                   !remove gravity effect
    SFEDELE,ALL,ALL,ALL
    SFE,E_Cable_%I%,,PRES,,-1                !apply unit elongation on cable
    SOLVE
*ENDDO
```

19.4.4 Postprocess 1

```
! POSTPROCESSING
!---------------------------------------------------------
! 1. calculate the cable forces using 'K stiffness matrix'
/POST1

*DIM,KMat,,NCables,NCables  !K matrix
*DIM,UYVect,,NCables        !UY vector  -   vertical deflection
*DIM,DLInitVect,,NCables    !Initial DL vector  -   cable shortening

!get and store the vertical deflection of the deck due to gravity
        SET,1
        *DO,I,1,NCables
            UYVect(I)=UY(DeckNodes(I))
        *ENDDO

!get and store the vertical deflection of the deck due to the unit cable forces
        *DO,ICable,1,NCables
            SET,1+ICable
            *DO,INode,1,NCables
                KMat(INode,ICable)=UY(DeckNodes(INode))
            *ENDDO
        *ENDDO

!solve the linear system to get cable forces at the final stage of the construction
        *MOPER,DLInitVect,KMat,SOLV,UYVect
```

19.4.5 Solution 2

```
! SOLUTION
!--------------------------------------------------------
! 2. the demolishing process - nonlinear analysis
/SOLU
ANTYPE,STATIC
NLGEOM,ON
NROPT,FULL
OUTRES,ALL,ALL

SFEDELE,ALL,ALL,ALL

!additional dead load (Gs)
CMSEL,S,E_Deck_All
SFBEAM,ALL,2,PRES,GsLoad

ALLSEL

! Apply cable forces (shortening)
*DO,I,1,NCables
    SFE,E_Cable_%I%,,PRES,,DLInitVect(I)
*ENDDO

ACEL,,9.81,
!LS1 - final state of the bridge
SOLVE
NLS=1

!LS2 - remove additonal dead load
SFEDELE,E_Deck_All,ALL,ALL
SOLVE
NLS=NLS+1

!LS3 - remove ROTZ bondary condition ("cut the bridge into two")
NSEL,S,LOC,X,(NCables+0.5)*XDistCable
DDELE,ALL,ALL
ALLSEL
SOLVE
NLS=NLS+1

!LS4 - remove the last deck section (no cable is attached to it)
EKILL,E_Deck_%NPhases%
SOLVE
NLS=NLS+1

!LS5 --> LS4+(Ncables*2)
*DO,I,NPhases-1,1,-1
    !kill cable
    EKILL,E_Cable_%I%
    SOLVE
    NLS=NLS+1

    !kill deck
    EKILL,E_Deck_%I%
    SOLVE
    NLS=NLS+1

*ENDDO
```

19.4.6 Postprocess 2

```
! POSTPROCESSING
```

```
!---------------------------------------------------------
/POST1

/ESHAPE,0
IStep=1

!element tables
*DO,I,1,NLS
    SET,I
    !cable forces
    ETABLE,CF_%I%,SMISC, 1
    !applied cable shortening
    ETAB,CST_%I%,NMISC, 3
*ENDDO

!combinations:
~TRGDEF,1,CROSS,M,Z,MIN
~TRGDEF,2,CROSS,M,Z,MAX
/TITLE,Construction
~CMBDEF,1,OPTION,NLS-1,0,1
*DO,I,1,NLS-1
~STSTDEF,1,I,LSTEP,I,0,1,1,1,1
*ENDDO
~COMBINE
```

19.4.7 Postprocess 3 (POST26)

```
! POSTPROCESSING
!---------------------------------------------------------
/POST26
! Change of cable forces
*DO,ICable,1,NCables
  ESOL,ICable+1,CableData(ICable,1),DeckNodes(ICable),SMISC,1,CForce_%ICable%
*ENDDO
```

19.4.8 Macros for postprocessing

File name: Example19_macros.inp

```
TO_START
IStep=NLS-1
/EOF

TO_END
IStep=1
/EOF

NEXT
*IF,IStep,GT,1,THEN
    IStep=IStep-1
*ENDIF
/EOF

BACK
*IF,IStep,LT,NLS-1,THEN
    IStep=IStep+1
*ENDIF
/EOF
```

```
UY
!vertical deflection
/POST1
*USE,ELEMSEL
SET,IStep
/PLOPTS,INFO,3
/PLOPTS,LEG2,0
/DSCALE,ALL,AUTO
/EFACET,1
PLNSOL, U,Y, 0,1.0
/EOF

MOMENTZ
!bending moment
/POST1
*USE,ELEMSEL
~CMBDAT,1
~CFSET,0,IStep,LAST,
/PLOPTS,INFO,2
/PLOPTS,LEG2,0
~PLLSFOR,M,Z,-1,
/EOF

FORCEX
!axial force
/POST1
*USE,ELEMSEL
~CMBDAT,1
~CFSET,0,IStep,LAST,
/PLOPTS,INFO,2
/PLOPTS,LEG2,0
~PLLSFOR,F,X,-1,
/EOF

CFORCE
!cable forces
/POST1
*USE,ELEMSEL
ESEL,R,TYPE,,3
/PLOPTS,INFO,3
/PLOPTS,LEG2,0
PLETAB,CF_%ARG1%,NOAV
/EOF

CSTROKE
!applied cable shortening
/POST1
*USE,ELEMSEL
ESEL,R,TYPE,,3
/PLOPTS,INFO,3
/PLOPTS,LEG2,0
PLETAB,CST_%ARG1%,NOAV
/EOF

CFORCE_GRAPH
/POST26
ALLSEL
*DO,ICable,1,NCables
  ESOL,ICable+1,CableData(ICable,1),DeckNodes(ICable),SMISC,1,CForce_%ICable%
*ENDDO
PLVAR,2,3,4,5,6,7,8,9,10,11
/EOF

ENVELOPE
/POST1
ALLSEL
~CMBDAT,2
```

```
*IF,ARG1,EQ,1,THEN
    ~CMB,1,CROSS,M,Z,MIN
*ELSE
    ~CMB,1,CROSS,M,Z,MAX
*ENDIF
/PLOPTS,INFO,2
/PLOPTS,LEG2,0
~PLLSFOR,M,Z,-1,
/EOF

ELEMSEL
!select elements that are alive during the loadstep
ESEL,NONE
*IF,ISTEP,LT,4,THEN
    ALLSEL
*ELSE
    *DO,I,1,NLS-IStep
        *IF,MOD(I,2),EQ,1,THEN
            CMSEL,A,E_DECK_%(I+1)/2%
        *ELSE
            CMSEL,A,E_Cable_%I/2%
        *ENDIF
    *ENDDO
*ENDIF
```

19.5 RESULTS

19.5.1 Bending moment (MZ) in different phases of the construction

In the final state of the bridge (LS1):



After removing the additional deadload (LS2):



After removing the symmetry boundary condition (LS3):



After removing the last deck segment (LS4):



After removing the last cable (LS5):



19.5.2 Other results in the final state of the structure

Axial force in the deck:



Vertical deflection of the deck:



Cable stroke (shortening):



Cable force:



19.5.3 Envelope of bending moment (MZ)

Positive bending moment (MZ+):



Negative bending moment (MZ-):



19.5.4 Time history results (POST26)

Variation of cable forces during construction: